Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
NotebookOpen[wdir <> "\\MakeVSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization / Utilities

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$TℏD = 4; ℏ /: ℏ^d_. /; d > $TℏD := 0;
$TϵD = 2; (* Can't be ∞ at least because of Quesne. Can't be ≤
 1 at least because of the explicit ϵ^2 in S𝔻$g. *)
ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
  Collect[Normal@Series[ℰ /. {T_i_ → ⅇ^(ℏ t_i/2), T → ⅇ^(ℏ t/2)}, {ℏ, 0, $TℏD}], ℏ, Together]]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

Differential polynomials (DP):

Utils

```
DP_{α_→D_x_, β_→D_y_}[P_][λ_] :=
 Total[CoefficientRules[P, {α, β}] /. ({m_, n_} → c_) :→ c D[λ, {x, m}, {y, n}]]
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, CE, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  CE[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  U_i_[NCM[]] := U[];
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := CE[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
  (a_ * x_U) ** y_ := CE[a (x ** y)]; x_ ** (a_ * y_U) := CE[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List ⧴ l_null, {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. l_s_ ⧴ (l /. x_i_ ⧴ x_s));
    CE[Total[
       CoefficientRules[poly, vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
    ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) ⧴ c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
  S_i_[c_. * u_U] := CE[(c /. S_i[U, Centrals]) DeleteCases[u, _i] **
     U_i[NCM @@ Reverse@Cases[u, x_i ⧴ S@U@x]]];
 ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) :→ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /.  u_U :→ m[u]];)
```

Meta-Operations

QLImplementation

```
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
(S@CU@y = -CU@y; S@CU@a = -CU@a; S@CU@x = -CU@x;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.640625, {(28 t² γ⁴ + 116 t γ⁵ ϵ + 120 γ⁶ ϵ²) CU[y, y, y, x, x] +
   ≪23≫ + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Verifying that S is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y_1, a_1, x_1}},
 Table[HL@Simp[S_1[z1 ** z2] - S_1[z2] ** S_1[z1]],
  {z1, bas}, {z2, bas} ] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of S on products of triples:

```
With[{bas = CU /@ {y1, a1, x1}},
 Table[HL@Simp[z1 ** z2 ** z3 - S1@S1[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Representing $sl_2^{\gamma\epsilon}$

Crho

```
{Cρ@y, Cρ@a, Cρ@x} = {(0 0; ϵ 0), (γ 0; 0 0), (0 γ; 0 0)};
Cρ[e^ℰ_] := MatrixExp[Cρ[ℰ]];
Cρ[ℰ_] := ℰ /. {t → γ ϵ, CU[u___] :→ Dot[(1 0; 0 1), Sequence @@ (Cρ /@ {u})]}
```

Verifying that $C\rho$ represents CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cρ[z1 ** z2] == Cρ[z1].Cρ[z2] // Simplify // HL,
  {z1, bas}, {z2, bas} ] ]
```

```
{{True, True, True}, {True, True, True}, {True, True, True}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle / \mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right) / \hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ħ)]; (*T=SS[e^(ħ t/2)];*)
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] = (q - 1) QU@{y, x} + 𝕆QU[SS[(1 - T^2 e^(-2 ϵ a ħ)) / ħ], {a}];
(S@QU@y = 𝕆QU[SS[-T^-2 e^(ħ ϵ a) y], {a, y}];
 S@QU@a = -QU@a; S@QU@x = 𝕆QU[SS[-e^(ħ ϵ a) x], {a, x}];)
Si_[QU, Centrals] = {ti → -ti, Ti → Ti^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to \gamma\, QU[y], \{QU[y], QU[x]\} \to \left(t + \frac{t^2 \hbar}{2} + \frac{t^3 \hbar^2}{6} + \frac{t^4 \hbar^3}{24}\right) QU[\,] + \left(-2 \epsilon - 2 t \epsilon \hbar - t^2 \epsilon \hbar^2 - \frac{1}{3} t^3 \epsilon \hbar^3\right) QU[a] + \left(2 \epsilon^2 \hbar + 2 t \epsilon^2 \hbar^2 + t^2 \epsilon^2 \hbar^3\right) QU[a, a] + \left(-\gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x]\},$$

$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y], \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$

$$\{\{QU[x], QU[y]\} \to \left(-t - \frac{t^2 \hbar}{2} - \frac{t^3 \hbar^2}{6} - \frac{t^4 \hbar^3}{24}\right) QU[\,] + \left(2 \epsilon + 2 t \epsilon \hbar + t^2 \epsilon \hbar^2 + \frac{1}{3} t^3 \epsilon \hbar^3\right) QU[a] + \left(-2 \epsilon^2 \hbar - 2 t \epsilon^2 \hbar^2 - t^2 \epsilon^2 \hbar^3\right) QU[a, a] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) QU[y, x], \{QU[x], QU[a]\} \to -\gamma\, QU[x], \{QU[x], QU[x]\} \to 0\}\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs @ $TℏD=5, $TϵD=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp]
  }] // Timing
```

$$\{46.0313, \{\left(28 t^2 \gamma^4 + 116 t \gamma^5 \epsilon + 120 \gamma^6 \epsilon^2 + 28 t^3 \gamma^4 \hbar + \ll 5 \gg + 7 t^5 \gamma^4 \hbar^3 + \frac{361}{3} t^4 \gamma^5 \epsilon \hbar^3 + \frac{2495}{3} t^3 \gamma^6 \epsilon^2 \hbar^3\right) QU[y, \ll 3 \gg, x] + \ll 22 \gg, 0\}\}$$

Verifying that *S* is an anti-homomorphism on QU:

```
With[{bas = QU /@ {y₁, a₁, x₁}},
 Table[{z1, z2} → HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas} ] ]
```

$$\{\{\{QU[y_1], QU[y_1]\} \to 0, \{QU[y_1], QU[a_1]\} \to 0, \{QU[y_1], QU[x_1]\} \to 0\},$$
$$\{\{QU[a_1], QU[y_1]\} \to 0, \{QU[a_1], QU[a_1]\} \to 0, \{QU[a_1], QU[x_1]\} \to 0\},$$
$$\{\{QU[x_1], QU[y_1]\} \to 0, \{QU[x_1], QU[a_1]\} \to 0, \{QU[x_1], QU[x_1]\} \to 0\}\}$$

Verifying that $\mathrm{Lim}_{\hbar \to 0}\, QU = CU$ using a “random” product (~23 secs @ $TℏD=5, $TϵD=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ℯ^(ℏ t/2)}, ℏ → 0] - lhs] // HL
  }] // Timing
```

$\{27.8594, \{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\,\mathrm{CU}[y, y, y, x, x] + (8\,t\,\gamma^5\,\epsilon + 16\,\gamma^6\,\epsilon^2)\,\mathrm{CU}[y, y, y, x, x] + \ll 106 \gg + \mathrm{CU}[y, y, y, y, y, a, a, a, a, x, x, x, x],$
$2\,\left(8\,t\,\gamma^6\,\epsilon^2\,\hbar + 12\,t^2\,\gamma^6\,\epsilon^2\,\hbar^2 + \frac{28}{3}\,t^3\,\gamma^6\,\epsilon^2\,\hbar^3\right)\,\mathrm{QU}[y, y, y, x, x] + \ll 566 \gg +$
$\left(\gamma\,\epsilon\,\hbar + \frac{15}{2}\,\gamma^2\,\epsilon^2\,\hbar^2\right)\,\mathrm{QU}[y, y, y, \ll 7 \gg, x, x, x], 0\}\}$

Representing $\mathcal{U}_{\gamma\epsilon;\hbar}$

Qrho

```
{Qρ@y, Qρ@a, Qρ@x} = {(0 0; ϵ 0), (γ 0; 0 0), (0 SS@((1-e^(-γ ϵ ℏ))/(ϵ ℏ)); 0 0)};
Qρ[ℰ_] := ℰ /. {t → γ ϵ, QU[u___] :→ Dot[(1 0; 0 1), Sequence @@ (Qρ /@ {u})]]}
```

Verifying that Cρ represents CU:

```
With[{bas = QU /@ {y, a, x}},
 Table[Qρ[z1 ** z2] == Qρ[z1].Qρ[z2] // Simplify // HL,
  {z1, bas}, {z2, bas}] ]
```

{{True, True, True}, {True, True, True}, {True, True, True}}

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ℯ^(ℏ ϵ a) x], {a, x}],
   a → -QU@a, x → 𝕆_QU[SS[-T^-1 ℯ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\left\{QU[y] \to \left(-1+\frac{t\hbar}{2}-\frac{t^2\hbar^2}{8}+\frac{t^3\hbar^3}{48}\right)QU[x]+\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3\right)QU[a,x]+\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3\right)QU[a,a,x]\to QU[y],\ QU[a]\to -QU[a]\to QU[a],\right.$$
$$QU[x]\to\left(-1+\frac{t\hbar}{2}+\gamma\epsilon\hbar-\frac{t^2\hbar^2}{8}-\frac{1}{2}t\gamma\epsilon\hbar^2-\frac{1}{2}\gamma^2\epsilon^2\hbar^2+\frac{t^3\hbar^3}{48}+\frac{1}{8}t^2\gamma\epsilon\hbar^3+\frac{1}{4}t\gamma^2\epsilon^2\hbar^3\right)QU[y]+$$
$$\left(-\epsilon\hbar+\frac{1}{2}t\epsilon\hbar^2+\gamma\epsilon^2\hbar^2-\frac{1}{8}t^2\epsilon\hbar^3-\frac{1}{2}t\gamma\epsilon^2\hbar^3\right)QU[y,a]+$$
$$\left.\left(-\frac{1}{2}\epsilon^2\hbar^2+\frac{1}{4}t\epsilon^2\hbar^3\right)QU[y,a,a]\to QU[x]\right\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\mathbb{AD}\$f = \frac{\gamma}{\hbar}e^{\hbar\left(\frac{t}{2}-(a+\gamma)\epsilon\right)}\left(\left(\mathrm{Cosh}\left[\hbar\left(a\epsilon+\frac{\gamma\epsilon}{2}-\frac{t}{2}\right)\right]-\mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\epsilon}{2}\right)^2+\epsilon\omega}\right]\right)\Big/\left(\mathrm{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(a^2\epsilon+a\gamma\epsilon-a\,t-\omega\right)\right)\right);$$

Scaling behaviour of $\mathbb{AD}\$f$:

```
HL@Simplify[AD$f == ((AD$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

True

```
HL@FullSimplify[
  AD$f == ((AD$f /. γ → 1) /. {ħ → γ^2 ħ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

True

ADeq

```
AD$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[AD, QU → CU,
  {a → CU@a, x → CU@x, y → S_CU[SS[AD$f], a → CU[a], ω → AD$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[AD[z1 ** z2] - AD[z1] ** AD[z2]]], {z1, bas}, {z2, bas}]]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$\text{SD\$g} = \sqrt{\left(\left(\text{Cosh}\left[\frac{\hbar}{2}\sqrt{t^2+\gamma^2\epsilon^2+4\epsilon\varpi}\right] - \text{Cosh}\left[\frac{\hbar}{2}\left(t-(2a+\gamma)\epsilon\right)\right]\right)\Big/\left(\text{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(t(2a+\gamma)-2a(a+\gamma)\epsilon+2\varpi\right)\hbar\big/(2\gamma)\right)\right)};$$

Verify agreement with the formulas in pensieve://People/VanDerVeen/Dequant1.pdf:

$$\Bigg\{\text{SD\$P} = \frac{\text{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon a\right)\right] - \text{Cosh}\left[\hbar\sqrt{\frac{t^2+\epsilon^2}{4}+\epsilon\varpi}\right]}{\hbar\,\text{Sinh}\left[\frac{-\epsilon\hbar}{2}\right]\left(\varpi-\epsilon a^2+(t-\epsilon)a+t/2\right)},$$

```
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
     SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 Simplify[SD$g == FullSimplify[
     √SD$Q /. c → a + 1/2 /. {ħ → γ^2 ħ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\Bigg\{-\left(\left(\left(\text{Cosh}\left[\left(a\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\text{Cosh}\left[\sqrt{\frac{1}{4}\left(t^2+\epsilon^2\right)+\epsilon\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a(t-\epsilon)-a^2\epsilon+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},$$

$$-\left(\left(4\left(\text{Cosh}\left[\frac{1}{2}(t-2c\epsilon)\hbar\right]-\text{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\epsilon\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\hbar}{2}\right]\right)\Big/\left(\left(4ct+\epsilon-4c^2\epsilon+4\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True}\Bigg\}$$

SDeq

```
S𝔻$f = FullSimplify[ℯ^(ħ (t/2-ϵ a)) (S𝔻$g /. {a → -a, t → -t})];
```

SDeq

```
S𝔻$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[S𝔻, QU → CU, {a → CU@a,
  x → 𝕊_CU[SS[S𝔻$f], a → CU[a], ϖ → S𝔻$ϖ] ** CU@x,
  y → 𝕊_CU[SS[S𝔻$g] , a → CU[a], ϖ → S𝔻$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[S𝔻[z]] == S𝔻[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[S𝔻[z1 ** z2] - S𝔻[z1] ** S𝔻[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

R in QU.

Quesne's formula:

Quesne

$$𝕖_{q\_,n\_}[x\_] := \mathrm{Exp}\left[\sum_{k=1}^{n} \frac{(1-q)^k}{k\,(1-q^k)} x^k\right];\quad 𝕖_{q\_}[x\_] := 𝕖_{q,\$T\epsilon D}[x]$$

```
Table[Together@SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)\,(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2\,(1+\rho^2)\,(1+\rho+\rho^2)},\right.$$
$$\left. 1\Big/\left((1+\rho)^2\,(1+\rho^2)\,(1+\rho+\rho^2)\,(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

```
{1, 1, 1, 1, 1, 1}
```

R

$$\mathrm{QU}[R_{i\_,j\_}] := \mathbb{O}_{QU}\left[\mathrm{SS}\left[e^{\hbar\, b_1 a_2}\, 𝕖_q[\hbar\, y_1 x_2] \;/.\; b_1 \to \gamma^{-1}\,(\epsilon\, a_1 - t_i)\right],\ \{y_1, a_1\}_i,\ \{a_2, x_2\}_j\right];$$
$$\mathrm{QU}\left[R^{-1}_{i\_,j\_}\right] := S_j\text{@}\mathrm{QU}[R_{i,j}];$$

```
QU[R3,4] // Short
```

$$\mathrm{QU}[\,] + \frac{\epsilon\hbar\,\mathrm{QU}[a_3, a_4]}{\gamma} + \hbar\,\mathrm{QU}[y_3, x_4] + \ll 20 \gg + \frac{\hbar^3\,\mathrm{QU}[y_3, a_4, a_4, x_4]\,t_3^2}{2\gamma^2} - \frac{\hbar^3\,\mathrm{QU}[a_4, a_4, a_4]\,t_3^3}{6\gamma^3}$$

Verifying R2 (~2 secs @ $TħD=4, $TϵD=2):

$\mathrm{QU}\left[R_{1,2} ** R^{-1}_{1,2}\right]$ `// Simp // HL // Timing`

{0.328125, QU[]}

Verifying R3 (~156 secs @ $TħD=4, $TϵD=2):

$\{\mathrm{Short}[\mathrm{lhs} = \mathrm{QU}[R_{1,2} ** R_{1,3} ** R_{2,3}]],\ \mathrm{HL@Simp}[\mathrm{lhs} - \mathrm{QU}[R_{2,3} ** R_{1,3} ** R_{1,2}]]\}$ `// Timing`

$$\left\{2.45313, \left\{\mathrm{QU}[\,] + \frac{\epsilon\hbar\,\mathrm{QU}[a_1, a_2]}{\gamma} + \ll 347 \gg + \mathrm{QU}[y_1, a_2, x_3]\left(2\epsilon\hbar^2 + 2\epsilon\hbar^3 t_2\right) + \mathrm{QU}[y_1, x_3]\left(-\hbar^2 t_2 - \frac{1}{2}\hbar^3 t_2^2\right), 0\right\}\right\}$$

The Classical Logos CΛ

Lemma 3C. To degree k,
$\mathbb{O}_{CU}\left(e^{\eta y+\xi x+\delta y x} \mid x y\right) = \mathbb{O}_{CU}\left(v\, e^{v(-t\xi\eta+\eta y+\xi x+\delta y x)}\, C\Lambda_k(\epsilon, \gamma, y, a, x, \eta, \xi, \delta) \mid y a x\right)$, with $v = (1 + t\delta)^{-1}$ and where $C\Lambda_k(\epsilon, \gamma, y, a, x, \eta, \xi, \delta)$ is a fixed polynomial of degree at most $4k$ in y, $\sqrt{a}$, x, η, ξ, with scalar coefficients.

Comment. Even better, $\log(C\Lambda_k)$ is of degree at most $2k+2$ in said variables.

$\mathrm{eqn} = C\rho\left[e^{\xi\,\mathrm{CU@x}}\right].C\rho\left[e^{\eta\,\mathrm{CU@y}}\right] == C\rho\left[e^{d\,\mathrm{CU@y}}\right].C\rho\left[e^{c\,(t\,\mathrm{CU}[\,] - 2\epsilon\,\mathrm{CU@a})}\right].C\rho\left[e^{b\,\mathrm{CU@x}}\right]$

$\{\{1 + \gamma\epsilon\eta\xi, \gamma\xi\}, \{\epsilon\eta, 1\}\} == \left\{\left\{e^{-c\gamma\epsilon}, b\, e^{-c\gamma\epsilon}\gamma\right\}, \left\{d\, e^{-c\gamma\epsilon}\epsilon, e^{c\gamma\epsilon} + b\, d\, e^{-c\gamma\epsilon}\gamma\epsilon\right\}\right\}$

`sol = Solve[Thread[Flatten /@ eqn], {d, b, c}][[1]] /. C[1] → 0`

$$\left\{d \to \frac{\eta}{1 + \gamma\epsilon\eta\xi},\ b \to \frac{\xi}{1 + \gamma\epsilon\eta\xi},\ c \to \frac{\mathrm{Log}\left[\frac{1}{1+\gamma\epsilon\eta\xi}\right]}{\gamma\epsilon}\right\}$$

Proof of Lemma 3C. We know that $\mathbb{O}_{CU}\left(e^{\xi x+\eta y} \mid x y\right) = \mathbb{O}_{CU}\left(e^{c t + a y - 2\epsilon c a + b x} \mid y a x\right)$, with $\left\{d \to \frac{\eta}{1+\gamma\epsilon\eta\xi}, b \to \frac{\xi}{1+\gamma\epsilon\eta\xi}, c \to \frac{\mathrm{Log}[1+\gamma\epsilon\eta\xi]}{-\gamma\epsilon}\right\}$. Expanding in ϵ we get
$\mathbb{O}_{CU}\left(e^{\xi x+\eta y} \mid x y\right) = \mathbb{O}_{CU}\left(\lambda_\epsilon(\xi, \eta)\, e^{\eta y + \xi x - \eta\xi t} \mid y a x\right) = \mathbb{O}_{CU}\left(\lambda_\epsilon(\partial_x, \partial_y)\, e^{\eta y + \xi x - \eta\xi t} \mid y a x\right)$ and so
$\mathbb{O}_{CU}\left(e^{\eta y+\xi x+\delta y x} \mid x y\right) = \mathbb{O}\left(\lambda_\epsilon(\partial_x, \partial_y)\, e^{\delta\partial_\eta\partial_\xi}\, e^{\eta y + \xi x - \eta\xi t} \mid y a x\right) = \mathbb{O}\left(\lambda_\epsilon(\partial_x, \partial_y)\, v\, e^{v(-t\xi\eta+\eta y+\xi x+\delta y x)} \mid y a x\right)$.

Logos

```
SSϵ[ℰ_] := Block[{ϵ}, Collect[Normal@Series[ℰ, {ϵ, 0, $TϵD}], ϵ, Together]];
(* Shielded ϵ-Series *)
CΛ[t_, y_, a_, x_, ξ1_, η1_, δ_] := Module[
  {eqn, d, b, c, sol, λ, q, ν, ξ, η},
  eqn = Cρ[𝕖^(ξ CU@x)].Cρ[𝕖^(η CU@y)] == Cρ[𝕖^(d CU@y)].Cρ[𝕖^(c (t CU[] - 2 ϵ CU@a))].Cρ[𝕖^(b CU@x)];
  sol = Solve[Thread[Flatten /@ eqn], {d, b, c}][[1]] /. C[1] → 0;
  λ = Simplify[𝕖^(-η y - ξ x + η ξ t) SSϵ[𝕖^(c t + d y - 2 ϵ c a + b x) /. sol]];
  q = 𝕖^(ν (-t ξ η + η y + ξ x + δ y x));
  Collect[q^-1 DP_{ξ→D_x, η→D_y}[λ][q] /. ν → (1 + t δ)^-1, ϵ, Simplify] /. {ξ → ξ1, η → η1}
 ];
```

```
CΛ[t, y, a, x, ξ, η, δ]
```

$$1+\frac{1}{24\,(1+t\,\delta)^8}\,\epsilon^2\Big(48\,a^2\,(1+t\,\delta)^4\,\big(2\,\delta^2\,(1+t\,\delta)^2+4\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)-$$
$$24\,a\,\gamma\,(1+t\,\delta)^4\,\big(2\,\delta^2\,(1+t\,\delta)^2+4\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)-$$
$$48\,a\,y\,\gamma\,(1+t\,\delta)^3\,(x\,\delta+\eta)$$
$$\big(6\,\delta^2\,(1+t\,\delta)^2+6\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)+24\,y\,\gamma^2\,(1+t\,\delta)^3$$
$$(x\,\delta+\eta)\,\big(6\,\delta^2\,(1+t\,\delta)^2+6\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)-48\,a\,x\,\gamma$$
$$(1+t\,\delta)^3\,(y\,\delta+\xi)\,\big(6\,\delta^2\,(1+t\,\delta)^2+6\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)+$$
$$24\,x\,\gamma^2\,(1+t\,\delta)^3\,(y\,\delta+\xi)\,\big(6\,\delta^2\,(1+t\,\delta)^2+6\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)+12\,y^2\,\gamma^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)^2$$
$$\big(12\,\delta^2\,(1+t\,\delta)^2+8\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)+12\,x^2\,\gamma^2$$
$$(1+t\,\delta)^2\,(y\,\delta+\xi)^2\,\big(12\,\delta^2\,(1+t\,\delta)^2+8\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)\,(y\,\delta+\xi)+(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2\big)+$$
$$24\,a\,t\,\gamma\,(1+t\,\delta)^2\,\big(6\,\delta^3\,(1+t\,\delta)^3+18\,\delta^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$9\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3\big)-$$
$$8\,t\,(\gamma+t\,\gamma\,\delta)^2\,\big(6\,\delta^3\,(1+t\,\delta)^3+18\,\delta^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$9\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3\big)+$$
$$24\,x\,y\,(\gamma+t\,\gamma\,\delta)^2\,\big(6\,\delta^3\,(1+t\,\delta)^3+18\,\delta^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$9\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3\big)-$$
$$12\,t\,y\,\gamma^2\,(1+t\,\delta)\,(x\,\delta+\eta)\,\big(24\,\delta^3\,(1+t\,\delta)^3+36\,\delta^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$12\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3\big)-$$
$$12\,t\,x\,\gamma^2\,(1+t\,\delta)\,(y\,\delta+\xi)\,\big(24\,\delta^3\,(1+t\,\delta)^3+36\,\delta^2\,(1+t\,\delta)^2\,(x\,\delta+\eta)\,(y\,\delta+\xi)+$$
$$12\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3\big)+$$
$$3\,t^2\,\gamma^2\,\big(24\,\delta^4\,(1+t\,\delta)^4+96\,\delta^3\,(1+t\,\delta)^3\,(x\,\delta+\eta)\,(y\,\delta+\xi)+72\,\delta^2\,(1+t\,\delta)^2$$
$$(x\,\delta+\eta)^2\,(y\,\delta+\xi)^2+16\,\delta\,(1+t\,\delta)\,(x\,\delta+\eta)^3\,(y\,\delta+\xi)^3+(x\,\delta+\eta)^4\,(y\,\delta+\xi)^4\big)\Big)+$$
$$\frac{1}{2\,(1+t\,\delta)^4}\,\epsilon\,\Big(4\,a\,(1+t\,\delta)^2\,\big((t+x\,y)\,\delta^2+\eta\,\xi+\delta\,(1+y\,\eta+x\,\xi)\big)+$$
$$\gamma\,\Big(2\,t^3\,\delta^4+4\,t^2\,\delta^2\,(\delta-x\,y\,\delta^2+\eta\,\xi)-2\,\big(y\,\eta\,(\delta\,(2+y\,\eta)+\eta\,\xi)+x^2\,\delta\,(2\,y^2\,\delta^2+3\,y\,\delta\,\xi+\xi^2)+$$
$$x\,(3\,y^2\,\delta^2\,\eta+4\,y\,\delta\,(\delta+\eta\,\xi)+\xi\,(2\,\delta+\eta\,\xi))\big)-t\,\big(3\,x^2\,y^2\,\delta^4-4\,\delta\,\eta\,\xi-\eta^2\,\xi^2+$$
$$4\,x\,y\,\delta^3\,(3+y\,\eta+x\,\xi)+\delta^2\,(-2+y^2\,\eta^2+4\,x\,\xi+x^2\,\xi^2+4\,y\,(\eta+x\,\eta\,\xi))\big)\Big)\Big)$$

```
{Short[lhs = 𝕆_CU[SS[e^(ℏ (ξ x+η y+δ x y))], {x, y}], 5], HL[lhs ==
   𝕆_CU[SS[ν e^(ℏ ν (ξ x+η y+δ x y-t ℏ ξ η)) CΛ[t, y, a, x, ℏ ξ, ℏ η, ℏ δ] /. ν → (1 + ℏ t δ)^-1], {y, a, x}]]}
```

$\{(1 - t\delta\hbar + t^2\delta^2\hbar^2 + t\gamma\delta^2\epsilon\hbar^2 - t\eta\xi\hbar^2 - t^3\delta^3\hbar^3 - 3t^2\gamma\delta^3\epsilon\hbar^3 - 2t\gamma^2\delta^3\epsilon^2\hbar^3 + 2t^2\delta\eta\xi\hbar^3 + 2t\gamma\delta\epsilon\eta\xi\hbar^3 + t^4\delta^4\hbar^4 + 6t^3\gamma\delta^4\epsilon\hbar^4 + 11t^2\gamma^2\delta^4\epsilon^2\hbar^4 - 3t^3\delta^2\eta\xi\hbar^4 - 9t^2\gamma\delta^2\epsilon\eta\xi\hbar^4 - 6t\gamma^2\delta^2\epsilon^2\eta\xi\hbar^4 + \frac{1}{2}t^2\eta^2\xi^2\hbar^4 + \frac{1}{2}t\gamma\epsilon\eta^2\xi^2\hbar^4)\,\mathrm{CU}[\,] + \ll 48 \gg + \frac{1}{24}\delta^4\hbar^4\,\mathrm{CU}[y, y, y, y, x, x, x, x],\ \mathrm{True}\}$

C𝕆 and Swaps

Swaps from Pensieve://Talks/Toulouse-1705/DogmaDemo.nb and from Pensieve://Talks/Sydney-1708/ExtraDetails@@.nb.

CdsO

```
SetAttributes[C𝕆, Orderless];
CU@C𝕆[specs___, 𝔼[ω_, L_, Q_, P_]] := 𝕆_CU[SS[ω e^(ℏ (L+Q)) P], specs]
```

```
CU@C𝕆[𝔼[1, t1 a2, t1^-1 (e^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}1, {x2, a2, y2}2] // Short
```

$\mathrm{CU}[\,] + \ll 27 \gg + \mathrm{CU}[y_1, x_1]\left(-\gamma\epsilon\hbar^2 t_2 + e^{t_1}\gamma\epsilon\hbar^2 t_2 + \frac{\epsilon\hbar \ll 1 \gg}{t_1} - \frac{\ll 1 \gg}{\ll 1 \gg} + \frac{1}{2}\gamma^2\epsilon\hbar^3 t_1 t_2 - \frac{1}{2}e^{t_1}\gamma^2\epsilon\hbar^3 t_1 t_2\right)$

```
HL[Cρ[e^(ξ CU@x)].Cρ[e^(α CU@a)] == Cρ[e^(α CU@a)].Cρ[e^(e^(-γ α) ξ CU@x)]]
```

True

SW

```
SW_{x_i_, a_j_}[C𝕆[{lh___, x_i_, a_j_, rh___}_s_, more___, 𝔼[ω_, L_, Q_, P_]]] :=
 C𝕆[{lh, a_j, x_i, rh}_s, more,
  With[{q = e^(-γ α) ξ x_i + α a_j},
   𝔼[ω, L, ℏ^-1 e^(-γ α) ξ x_i + (Q /. x_i → 0), e^-q DP_{x_i→D_ξ, a_j→D_α}[P][e^q]] /. {α → ℏ ∂_{a_j} L, ξ → ℏ ∂_{x_i} Q}]]
```

```
c𝕆 = C𝕆[𝔼[1, t1 a2, t1^-1 (e^t1 - 1) y1 x2, 1 + ϵ x1 y2], {y1, x1}1, {x2, a2, y2}2]
```

$\mathrm{C}\mathbb{O}\left[\{y_1, x_1\}_1, \{x_2, a_2, y_2\}_2, \mathbb{E}\left[1, a_2 t_1, \frac{(-1 + e^{t_1})\,x_2 y_1}{t_1}, 1 + \epsilon x_1 y_2\right]\right]$

```
SW_{x2,a2}[c𝕆]
```

$\mathrm{C}\mathbb{O}\left[\{y_1, x_1\}_1, \{a_2, x_2, y_2\}_2, \mathbb{E}\left[1, a_2 t_1, \frac{e^{-\gamma\hbar t_1}(-1 + e^{t_1})\,x_2 y_1}{t_1}, 1 + \epsilon x_1 y_2\right]\right]$

```
With[{c𝕆 = C𝕆[{y1, x1}1, {x2, a2, y2}2, 𝔼[1, t1 a2, t1^-1 (e^t1 - 1) y1 x2, 1 + ϵ x1 y2]]},
 HL[CU[c𝕆] == CU[c𝕆 // SW_{x2,a2}]]]
```

True

```
With[{c⊕ = C⊕[{y1, a1, x1}1, {x2, a2, y2}2,
    𝔼[ω, l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2, γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2,
     1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]]},
 {CU[c⊕] // Short, HL[CU[c⊕] == CU[c⊕ // SW_{x2,a2}]]}
]
```

$\{\mathrm{CU}[a_1, a_1, a_1, a_1]\left(\frac{1}{6}\epsilon\,\omega\,\hbar^3\,l_1\,l_{11}^3\,t_1^3 + \frac{1}{2}\epsilon\,\omega\,\hbar^3\,l_1\,l_{11}^2\,l_{21}\,t_1^2\,t_2 + \frac{1}{2} \ll 7 \gg \ll 1 \gg + \frac{1}{6}\epsilon\,\omega\,\hbar^3\,l_1\,l_{21}^3\,t_2^3\right) + \ll 180 \gg + \ll 1 \gg, \text{True}\}$

Stitching Direct

```
MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].
  MatrixExp[ξ1 Cρ[CU@x]].MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].
  MatrixExp[ξ2 Cρ[CU@x]] // Simplify // MatrixForm
```

$$\begin{pmatrix} e^{\gamma(\alpha_1+\alpha_2)}\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right) & e^{\gamma\alpha_1}\gamma\left(e^{\gamma\alpha_2}\xi_2+\xi_1\left(1+e^{\gamma\alpha_2}\gamma\,\epsilon\,\eta_2\,\xi_2\right)\right) \\ e^{\gamma\alpha_2}\epsilon\left(\eta_2+e^{\gamma\alpha_1}\eta_1\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right)\right) & 1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1+e^{\gamma\alpha_2}\gamma\,\epsilon\left(\eta_2+e^{\gamma\alpha_1}\eta_1\left(1+\gamma\,\epsilon\,\eta_2\,\xi_1\right)\right)\xi_2 \end{pmatrix}$$

```
eqn = MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].MatrixExp[ξ1 Cρ[CU@x]].
  MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].MatrixExp[ξ2 Cρ[CU@x]] ==
 e^(τ0 ϵ γ) MatrixExp[η0 Cρ[CU@y]].MatrixExp[α0 Cρ[CU@a]].MatrixExp[ξ0 Cρ[CU@x]]
```

$$\{\{e^{\gamma\alpha_2}\left(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\right),\ e^{\gamma\alpha_1}\gamma\,\xi_1+e^{\gamma\alpha_2}\gamma\left(e^{\gamma\alpha_1}+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_2\,\xi_1\right)\xi_2\},$$
$$\{e^{\gamma\alpha_2}\left(e^{\gamma\alpha_1}\epsilon\,\eta_1+\epsilon\,\eta_2\left(1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right),$$
$$1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1+e^{\gamma\alpha_2}\gamma\left(e^{\gamma\alpha_1}\epsilon\,\eta_1+\epsilon\,\eta_2\left(1+e^{\gamma\alpha_1}\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right)\xi_2\}\} ==$$
$$\{\{e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0},\ e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0}\gamma\,\xi0\},\ \{e^{\alpha0\,\gamma+\gamma\,\epsilon\,\tau0}\epsilon\,\eta0,\ e^{\gamma\,\epsilon\,\tau0}\left(1+e^{\alpha0\,\gamma}\gamma\,\epsilon\,\eta0\,\xi0\right)\}\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {τ0, η0, α0, ξ0}]][[1]]
```

Solve: Inconsistent or redundant transcendental equation. After reduction, the bad equation is

$\mathrm{Log}[e^{\gamma(\alpha 0+\epsilon\,\tau 0)}] - \mathrm{Log}[e^{\gamma\,\alpha_2}(e^{\gamma\,\mathrm{Subscript}[\ll 2\gg]} + e^{\mathrm{Times}[\ll 2\gg]}\,\gamma\,\epsilon\,\eta_2\,\xi_1)] == 0.$

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Solve: Equations may not give solutions for all "solve" variables.

$$\Big\{\tau 0 \to \frac{1}{\gamma\,\epsilon}\left(-\mathrm{Log}\left[e^{\alpha 0\,\gamma}\right] + \mathrm{Log}\left[e^{\gamma\,\alpha_1+\gamma\,\alpha_2} + e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_1\right]\right),\ \eta 0 \to \frac{1}{\gamma\,\epsilon\,\left(\xi_1 + e^{\gamma\,\alpha_2}\,\xi_2 + e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_1\,\xi_2\right)}$$
$$e^{-\gamma\,\alpha_1}\Bigg(\frac{1}{2} + \frac{1}{2}e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2 -$$
$$\frac{1}{2}\sqrt{\Big(\left(-1 - e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2\right)^2 +}$$
$$4\,e^{-\alpha 0\,\gamma+\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\Big(-e^{\gamma\,\alpha_1}\,\eta_1\,\xi_1 - \eta_2\,\xi_1 - e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1^2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\eta_2\,\xi_2 -$$
$$2\,e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2^2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma^2\,\epsilon^2\,\eta_1\,\eta_2^2\,\xi_1^2\,\xi_2\Big)\Big)\Bigg),$$
$$\xi 0 \to \frac{1}{e^{\gamma\,\alpha_1}\,\eta_1 + \eta_2 + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1}\,e^{-\gamma\,\alpha_2}\Bigg(\frac{1}{2\,\gamma\,\epsilon} + \frac{1}{2}e^{\gamma\,\alpha_1}\,\eta_1\,\xi_1 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\eta_1\,\xi_2 +$$
$$\frac{1}{2}e^{\gamma\,\alpha_2}\,\eta_2\,\xi_2 + \frac{1}{2}e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 -$$
$$\frac{1}{2\,\gamma\,\epsilon}\Big(\sqrt{\Big(\left(-1 - e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma^2\,\epsilon^2\,\eta_1\,\eta_2\,\xi_1\,\xi_2\right)^2 +}$$
$$4\,e^{-\alpha 0\,\gamma+\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\Big(-e^{\gamma\,\alpha_1}\,\eta_1\,\xi_1 - \eta_2\,\xi_1 - e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1^2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\eta_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\eta_2\,\xi_2 -$$
$$2\,e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_2}\,\gamma\,\epsilon\,\eta_2^2\,\xi_1\,\xi_2 - e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\,\gamma^2\,\epsilon^2\,\eta_1\,\eta_2^2\,\xi_1^2\,\xi_2\Big)\Big)\Big)\Bigg)\Big\}$$

```
eqn = MatrixExp[η1 Cρ[CU@y]].MatrixExp[α1 Cρ[CU@a]].MatrixExp[ξ1 Cρ[CU@x]].
   MatrixExp[η2 Cρ[CU@y]].MatrixExp[α2 Cρ[CU@a]].MatrixExp[ξ2 Cρ[CU@x]] ==
  T0 MatrixExp[η0 Cρ[CU@y]].MatrixExp[α0 Cρ[CU@a]].MatrixExp[ξ0 Cρ[CU@x]]
```

$$\Big\{\Big\{e^{\gamma\,\alpha_2}\left(e^{\gamma\,\alpha_1} + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_2\,\xi_1\right),\ e^{\gamma\,\alpha_1}\,\gamma\,\xi_1 + e^{\gamma\,\alpha_2}\,\gamma\left(e^{\gamma\,\alpha_1} + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_2\,\xi_1\right)\xi_2\Big\},$$
$$\Big\{e^{\gamma\,\alpha_2}\left(e^{\gamma\,\alpha_1}\,\epsilon\,\eta_1 + \epsilon\,\eta_2\left(1 + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right),$$
$$1 + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1 + e^{\gamma\,\alpha_2}\,\gamma\left(e^{\gamma\,\alpha_1}\,\epsilon\,\eta_1 + \epsilon\,\eta_2\left(1 + e^{\gamma\,\alpha_1}\,\gamma\,\epsilon\,\eta_1\,\xi_1\right)\right)\xi_2\Big\}\Big\} ==$$
$$\Big\{\Big\{e^{\alpha 0\,\gamma}\,T0,\ e^{\alpha 0\,\gamma}\,T0\,\gamma\,\xi 0\Big\},\ \Big\{e^{\alpha 0\,\gamma}\,T0\,\epsilon\,\eta 0,\ T0\left(1 + e^{\alpha 0\,\gamma}\,\gamma\,\epsilon\,\eta 0\,\xi 0\right)\Big\}\Big\}$$

```
sol = Block[{ϵ}, Solve[Thread[Flatten /@ eqn], {T0, η0, α0, ξ0}]][[1]]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

$$\Big\{T0 \to \frac{1}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1},\ \eta 0 \to \frac{\eta_1 + e^{-\gamma\,\alpha_1}\,\eta_2 + \gamma\,\epsilon\,\eta_1\,\eta_2\,\xi_1}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1},$$
$$\alpha 0 \to \frac{\mathrm{Log}\left[e^{\gamma\,\alpha_1+\gamma\,\alpha_2}\left(1 + \gamma\,\epsilon\,\eta_2\,\xi_1\right)^2\right]}{\gamma},\ \xi 0 \to \frac{e^{-\gamma\,\alpha_2}\,\xi_1 + \xi_2 + \gamma\,\epsilon\,\eta_2\,\xi_1\,\xi_2}{1 + \gamma\,\epsilon\,\eta_2\,\xi_1}\Big\}$$

𝔼

$\mathbb{E}[\omega, L, Q, P]$ means $\omega e^{\hbar(L+Q)} P$, where ω is a scalar, L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial. It should be interpreted via $\mathrm{C}\mathbb{O}[\mathbb{E}[\ldots],\ \{x_1,\ a_1,\ y_1\}_i,\ \ldots]$ (with some default for direct interpretation), or likewise via $\mathrm{Q}\mathbb{O}[\mathbb{E}[\ldots],\ \{x_1,\ a_1,\ y_1\}_i,\ \ldots]$. In themselves, $\mathrm{C}\mathbb{O}$ and $\mathrm{Q}\mathbb{O}$ should have an interpretation in CU/QU by casting.